Math 276
Calculus III

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 19

Double Integrals in Polar Coordinates

Initializations

```
> restart;
  with(plots):
  with(student):
>
```

19.1 Double Integrals in Polar Coordinates

As explained in class the double integral in Cartesian coordinates $\int_c^d \int_a^b f(x, y)\, dx\, dy$ can be translated into the following double integral in Polar coordinates: $\int_{\theta_1}^{\theta_2} \int_{r_1}^{r_2} f(r\cos(\theta), r\sin(\theta))\, r\, dr\, d\theta$.

Provided the integration regions are the same.

Examples

Example 19.1.1

Integrate the function $f(x, y) = 3x + 4y^2$ over the region in the upper half plane bounded by $x^2 + y^2 = 1$ and $x^2 + y^2 = 4$.

Solution

First we enter the data and sketch the integration region. When working with polar coordinates it is often helpful to incorporate the coordinate axes in the plot and have equal scaling on both axes.

```
> f:=(x, y)->3*x+4*y^2;
  bnd1:=x^2+y^2=1;
  bnd2:=x^2+y^2=4;
  bnd3:=y=0;
```

$$f := (x, y) \to 3x + 4y^2$$

$$bnd1 := x^2 + y^2 = 1$$

$$bnd2 := x^2 + y^2 = 4$$

$$bnd3 := y = 0 \qquad \textbf{(2.1.1.1)}$$

```
> implicitplot({seq(bnd||k, k=1..3)}, x=-3..3, y=-0.5..3,
```

```
scaling=constrained, axes=frame);
```



We can integrate over r from 1 to 2 and over θ from 0 to π.

```
> e1:=Doubleint(f(r*cos(theta), r*sin(theta))*r, r=1..2,
  theta=0..Pi);
  e2:=value(e1);
```

$$e1 := \int_0^\pi \int_1^2 \left(3\, r \cos(\theta) + 4\, r^2 \sin(\theta)^2\right) r \, dr \, d\theta$$

$$e2 := \frac{15}{2}\pi \qquad (2.1.1.2)$$

```
>
```

Example 19.1.2

Compute the volume of the solid above the cone $z = \sqrt{x^2 + y^2}$ and below the sphere $x^2 + y^2 + z^2 = 1$.

Solution

First define the functions and plot the footprint of the solid. It should be clear that we need the top half of the sphere.

```
> f:=(x, y)->sqrt(x^2+y^2);
  g:=(x, y)->sqrt(1-x^2-y^2);
```

$$f := (x, y) \rightarrow \sqrt{x^2 + y^2}$$

$$g := (x, y) \rightarrow \sqrt{1 - x^2 - y^2} \qquad (2.1.2.1)$$

Next, we determine where the surfaces intersect.

```
> bnd:=f(x, y)=g(x, y);
```

$$bnd := \sqrt{x^2+y^2} = \sqrt{1-x^2-y^2} \quad \textbf{(2.1.2.2)}$$

```
> bnd:=map(u->u^2, bnd);
```

$$bnd := x^2+y^2 = 1-x^2-y^2 \quad \textbf{(2.1.2.3)}$$

```
> bnd:=map(u->u+x^2+y^2, bnd);
```

$$bnd := 2\,x^2+2\,y^2 = 1 \quad \textbf{(2.1.2.4)}$$

```
> bnd:=bnd/2;
```

$$bnd := x^2+y^2 = \frac{1}{2} \quad \textbf{(2.1.2.5)}$$

Hence, the surfaces intersect on the circle $x^2+y^2=\frac{1}{2}$ in the plane $z=\frac{1}{2}\sqrt{2}$

```
> implicitplot(bnd, x=-1..1, y=-1..1, scaling=constrained);
```



From the plot and the equation of the boundary we derive that θ ranges between 0 and 2π, and r ranges from 0 to $\frac{1}{2}\sqrt{2}$. In order to simplify our coding, we will separately define the polar coordinates and substitute those into the Cartesian equations of the cone and the sphere.

```
> polar:={x=r*cos(theta), y=r*sin(theta)};
```

$$\text{polar} := \{x = r\cos(\theta), y = r\sin(\theta)\} \quad \textbf{(2.1.2.6)}$$

```
> e1:=Doubleint(subs(polar, g(x, y)-f(x, y))*r, theta=0..2*
  Pi, r=0..sqrt(1/2));
```

$$e1 := \int_0^{\frac{1}{2}\sqrt{2}} \int_0^{2\pi} \left(\sqrt{1-r^2\cos(\theta)^2-r^2\sin(\theta)^2} - \sqrt{r^2\cos(\theta)^2+r^2\sin(\theta)^2}\right) r \,\mathrm{d}\theta \,\mathrm{d}r \quad \textbf{(2.1.2.7)}$$

```
> e2:=simplify(e1, symbolic);
```

$$e2 := -\left(\int_0^{2\pi} 1 \,\mathrm{d}\theta\right)\left(\int_0^{\frac{1}{2}\sqrt{2}} \left(-\sqrt{1-r^2}+r\right) r \,\mathrm{d}r\right) \quad \textbf{(2.1.2.8)}$$

```
> e3:=simplify(value(e2));
  evalf(e3);
```

$$e3 := -\frac{1}{3}\pi\left(-2+\sqrt{2}\right)$$

$$0.6134341233 \quad (2.1.2.9)$$

For the true graphics enthusiast it may be satisfying to plot a picture of this solid.

```
> p1:=plot3d(subs(polar, [x, y, f(x, y)]), theta=Pi/3..2*
  Pi+Pi/6, r=0..1, style=patch):
  p2:=plot3d(subs(polar, [x, y, g(x, y)]), theta=4*Pi/10.
  .2*Pi+Pi/10, r=0..1, style=patch, color=red):
  p3:=plot3d(subs(polar, [x, y, 0]), theta=0..2*Pi, r=0..
  sqrt(1/2), color=[0.5, 0.5, 0.5], grid=[25, 5]):
  display([p1, p2, p3], axes=frame, scaling=constrained,
  orientation=[58, 64]);
```




```
>
```

The grey disk in the xy plane represents the integration region.

Example 19.1.3

Compute the area enclosed by one leaf of the rose $r = \cos(3\,\theta)$.

Solution

Now we know that the element of area $dA = dx\,dy$ translates as $dA = r\,dr\,d\,\theta$, we are able to express the area enclosed by a curve in Polar coordinates as a double rather than a single integral. The formula for the double integral may be easier to remember than the old single integral expression for the same area.

```
> f:=theta->cos(3*theta);
```

$$f := \theta \rightarrow \cos(3\,\theta) \quad (2.1.3.1)$$

```
> polarplot(f(theta), theta=0..Pi, color=magenta, scaling=
```

```
constrained, axes=boxed);
```




```
> area:=Doubleint(r, r=0..f(theta), theta=-Pi/6..Pi/6);
```

$$area := \int_{-\frac{1}{6}\pi}^{\frac{1}{6}\pi} \int_{0}^{\cos(3\theta)} r \, dr \, d\theta \tag{2.1.3.2}$$

```
> answer:=value(area);
```

$$answer := \frac{1}{12}\pi \tag{2.1.3.3}$$

```
>
```